A6.1

```
> eq[EW] := EW = (H+a)/2;
```

$$eq_{EW} := EW = \frac{1}{2} H + \frac{1}{2} a \tag{1.1}$$

```
> eq[EC] := EC = C/p;
```

$$eq_{EC} := EC = \frac{C}{p} \tag{1.2}$$

```
> eq[p] := p = (H-a)/(H-L);
```

$$eq_{p} := p = \frac{H - a}{H - L} \tag{1.3}$$

```
> eval(subs(eq[EW],eq[EC],eq[p], EW-EC));
```

$$\frac{1}{2} H + \frac{1}{2} a - \frac{C\,(H - L)}{H - a} \tag{1.4}$$

```
> Def[V] := V = unapply(%,a);
```

$$Def_{V} := V = \left(a \rightarrow \frac{1}{2} H + \frac{1}{2} a - \frac{C\,(H - L)}{H - a} \right) \tag{1.5}$$

```
> xmp := {H=4000,L=2000,C=360};
```

$$xmp := \{H = 4000, L = 2000, C = 360\} \tag{1.6}$$

```
> V(a)=a;
```

$$V(a) = a \tag{1.7}$$

```
> Opt := eval(subs(Def[V],xmp,%));
```

$$Opt := 2000 + \frac{1}{2} a - \frac{720000}{4000 - a} = a \tag{1.8}$$

```
> plot([lhs(Opt),rhs(Opt)],a=2000..3500,thickness=3);
```




```
> solve({Opt,a<subs(xmp,H)},{a});
```

$$\{a = 2800\} \tag{1.9}$$

```
>
>
>
>
```

A6.3

```
> eq[DU] := Delta[U] = p[EU]*E + p[OU]*O - p[UE]*U - p[UO]*U ;
> eq[DE] := Delta[E] = p[UE]*U + p[OE]*O - p[EU]*E - p[EO]*E ;
```

$$eq_{DU} := \Delta_U = p_{EU} E + p_{OU} O - p_{UE} U - p_{UO} U$$

$$eq_{DE} := \Delta_E = p_{UE} U + p_{OE} O - p_{EU} E - p_{EO} E \tag{2.1}$$

```
> Def[u] := u = U/(U+E);
```

(2.2)

$$Def_u := u = \frac{U}{U+E} \tag{2.2}$$

```
> eq[Pop] := N = U + E + O;
```

$$eq_{Pop} := N = U + E + O \tag{2.3}$$

```
> subs(Delta[U]=0,Delta[E]=0,{eq[DU],eq[DE],eq[Pop]});
```

$$\{N = U + E + O,\ 0 = p_{EU} E + p_{OU} O - p_{UE} U - p_{UO} U,\ 0 = p_{UE} U + p_{OE} O - p_{EU} E - p_{EO} E\} \tag{2.4}$$

```
> Sol[EOU] := solve(%,{E,O,U});
```

$$Sol_{EOU} := \left\{ E = \left(N \left(p_{UE} p_{OU} + p_{UE} p_{OE} + p_{UO} p_{OE}\right)\right) \Big/ \left(p_{OE} p_{EU} + p_{UE} p_{OU} + p_{OU} p_{EU} + p_{OU} p_{EO} + p_{UE} p_{OE} + p_{UE} p_{EO} + p_{UO} p_{EU} + p_{UO} p_{OE} + p_{UO} p_{EO}\right),\ U = \left(\left(p_{OE} p_{EU} + p_{OU} p_{EU} + p_{OU} p_{EO}\right) N\right) \Big/ \left(p_{OE} p_{EU} + p_{UE} p_{OU} + p_{OU} p_{EU} + p_{OU} p_{EO} + p_{UE} p_{OE} + p_{UE} p_{EO} + p_{UO} p_{EU} + p_{UO} p_{OE} + p_{UO} p_{EO}\right),\ O = \left(N \left(p_{UE} p_{EO} + p_{UO} p_{EU} + p_{UO} p_{EO}\right)\right) \Big/ \left(p_{OE} p_{EU} + p_{UE} p_{OU} + p_{OU} p_{EU} + p_{OU} p_{EO} + p_{UE} p_{OE} + p_{UE} p_{EO} + p_{UO} p_{EU} + p_{UO} p_{OE} + p_{UO} p_{EO}\right) \right\} \tag{2.5}$$

```
> Sol[u] := factor(eval(subs(Sol[EOU] , Def[u])));
```

$$Sol_u := u = \frac{p_{OE} p_{EU} + p_{OU} p_{EU} + p_{OU} p_{EO}}{p_{OE} p_{EU} + p_{OU} p_{EU} + p_{OU} p_{EO} + p_{UE} p_{OU} + p_{UE} p_{OE} + p_{UO} p_{OE}} \tag{2.6}$$

```
> collect(Sol[u],[p[EU],p[UE]],factor);
```

$$u = \frac{\left(p_{OE} + p_{OU}\right) p_{EU} + p_{OU} p_{EO}}{\left(p_{OE} + p_{OU}\right) p_{EU} + \left(p_{OE} + p_{OU}\right) p_{UE} + p_{OU} p_{EO} + p_{UO} p_{OE}} \tag{2.7}$$

```
> subs(p[EO]=0,limit(Sol[u],p[OE]=0));
```

$$u = \frac{p_{EU}}{p_{UE} + p_{EU}} \tag{2.8}$$

```
> Def[theta] := theta = p[OU]/(p[OE]+p[OU]);
```

$$Def_\theta := \theta = \frac{p_{OU}}{p_{OE} + p_{OU}} \tag{2.9}$$

```
> collect(eval(subs(solve(Def[theta],{p[OU]}),Sol[u])),theta,
  factor);
```

$$u = \frac{p_{EU} + \theta\, p_{EO}}{p_{EU} + \theta\, p_{EO} + p_{UE} - p_{UO}\, \theta + p_{UO}} \tag{2.10}$$

```
>
> xmp := { p[EU] = 2/100, p[EO] = 4/100, p[UE] = 20/100, p[UO]
  = 15/100, p[OE]=5/100,p[OU] = 3/100};
```

(2.11)

$$xmp := \left\{ p_{EU} = \frac{1}{50}, p_{EO} = \frac{1}{25}, p_{UE} = \frac{1}{5}, p_{UO} = \frac{3}{20}, p_{OE} = \frac{1}{20}, p_{OU} = \frac{3}{100} \right\} \quad (2.11)$$

```
> eval(subs(xmp,Sol[u]));
```

$$u = \frac{28}{263} \quad (2.12)$$

```
> evalf(%);
```

$$u = 0.1064638783 \quad (2.13)$$

>

A6.4

```
> restart;
> xmp := {p[A] = (3)/10, p[C] =(3)/10, p[B] = 2/10, w[A] =
  60000, w[C] = 80000, w[B] = 100000};
```

$$xmp := \left\{ p_A = \frac{3}{10}, p_C = \frac{3}{10}, p_B = \frac{1}{5}, w_A = 60000, w_C = 80000, w_B = 100000 \right\} \quad (3.1)$$

```
> eq[EW[A]] := W[A] = (p[A]*w[A] + p[B]*w[B] + p[C]*w[C] )/ (p
  [A] + p[B] + p[C] );
```

$$eq_{EW_A} := W_A = \frac{p_A w_A + p_B w_B + p_C w_C}{p_A + p_B + p_C} \quad (3.2)$$

```
> eq[EW[C]] := W[C] = ( p[B]*w[B] + p[C]*w[C] )/ (p[B] + p[C] )
  ;
```

$$eq_{EW_C} := W_C = \frac{p_B w_B + p_C w_C}{p_B + p_C} \quad (3.3)$$

```
> eq[EW[B]] := W[B] = w[B] ;
```

$$eq_{EW_B} := W_B = w_B \quad (3.4)$$

```
> eq[Ec[A]] := EC= c / (p[A] + p[B] + p[C] );
```

$$eq_{Ec_A} := EC = \frac{c}{p_A + p_B + p_C} \quad (3.5)$$

```
> eq[Ec[C]] := EC= c / ( p[B] + p[C] );
```

$$eq_{Ec_C} := EC = \frac{c}{p_B + p_C} \quad (3.6)$$

```
> eq[Ec[B]] := EC= c / ( p[B]  );
```

$$eq_{Ec_B} := EC = \frac{c}{p_B} \quad (3.7)$$

```
> W[A] - EC ;
```

$$W_A - EC \quad (3.8)$$

```
> eval(subs(eq[EW[A]], eq[Ec[A]] ,xmp,%));
```

$$77500 - \frac{5}{4}\,c \tag{3.9}$$

```
> % = w[A];
```

$$77500 - \frac{5}{4}\,c = w_A \tag{3.10}$$

```
> solve(subs(xmp,%),{c});
```

$$\{c = 14000\} \tag{3.11}$$

```
> eq[w[A]] := v[A] = eval(subs(eq[EW[A]], eq[Ec[A]] ,xmp,W[A] -
  EC ));
```

$$eq_{w_A} := v_A = 77500 - \frac{5}{4}\,c \tag{3.12}$$

```
> eq[w[B]] := v[B] = eval(subs(eq[EW[B]], eq[Ec[B]] ,xmp,W[B] -
  EC ));
```

$$eq_{w_B} := v_B = 100000 - 5\,c \tag{3.13}$$

```
> solve(subs(v[B]=w[B],xmp,%),{c});
```

$$\{c = 0\} \tag{3.14}$$

```
> eq[w[C]] := v[C] = eval(subs(eq[EW[C]], eq[Ec[C]] ,xmp,W[C] -
  EC ));
```

$$eq_{w_C} := v_C = 88000 - 2\,c \tag{3.15}$$

```
> solve(subs(v[C]=w[C],xmp,%),{c});
```

$$\{c = 4000\} \tag{3.16}$$

```
> Sol[a] := piecewise(c<4000,w[C],w[A]);
```

$$Sol_a := \begin{cases} w_C & c < 4000 \\ w_A & otherwise \end{cases} \tag{3.17}$$

```
> eval(subs(eq[EW[A]], eq[Ec[A]] ,xmp,W[A] - EC));
```

$$77500 - \frac{5}{4}\,c \tag{3.18}$$

```
> solve(%,{c});
```

$$\{c = 62000\} \tag{3.19}$$

If the cost is less than 4000, accept only job B (acceptance wage is 100000-5c > 80000)
If the cost is more than 4000, but less than 14000 accept B or C (but not A) (acceptance wage is 88000-2c > 60000)
If the cost is more than 14000, but less than 62000 accept any job
If the cost is more than 62000 stop searching

```
>
> eval(subs(eq[w[A]],eq[w[B]],eq[w[C]],max(v[A],v[B],v[C])));
> plot(subs(xmp,[%,w[A],w[C]]),c=0..62000,thickness=3,title=
  "Net value of search");
```

$$\max\left(77500 - \frac{5}{4}\,c,\ 100000 - 5\,c,\ 88000 - 2\,c\right)$$